

# Mathematical Programming: Modelling and Applications

Sonia Cafieri

LIX, École Polytechnique

`caferi@lix.polytechnique.fr`

October 2009

# Graph partitioning: AMPL model

```
param n >= 1, integer; # number of vertices
set V := 1..n;
set E within {V,V};

param c{E}; # edge weights

param kmax; # max number of clusters
set K := 1..kmax;

var x{V,K} binary; # original problem variables
var w{V,K,V,K} >= 0, <= 1; # linearization variables

minimize intercluster :
    sum{k in K, l in K, (u,v) in E : k != l} c[u,v] * w[u,k,v,l];

# constraints
subject to assignment {v in V} : sum{k in K} x[v,k] = 1;

subject to existence {k in K} : sum{v in V} x[v,k] >= 1;

# linearization constraints
subject to lin1 {u in V, v in V, h in K, k in K : (u,v) in E or (v,u) in E} :
    w[u,h,v,k] <= x[u,h];

subject to lin2 {u in V, v in V, h in K, k in K : (u,v) in E or (v,u) in E} :
    w[u,h,v,k] <= x[v,k];

subject to lin3 {u in V, v in V, h in K, k in K : (u,v) in E or (v,u) in E} :
    w[u,h,v,k] >= x[u,h] + x[v,k] - 1;
```

# Graph partitioning: Solution

```
ILOG AMPL 11.010, licensed to "ecolepolytechnique-palaiseau".
AMPL Version 20080219 (Linux 2.6.9-5.ELsmp)
ILOG CPLEX 10.100, licensed to "ecolepolytechnique-palaiseau", options: e m b q use=:
CPLEX 10.1.0: optimal integer solution; objective 3
1850 MIP simplex iterations
149 branch-and-bound nodes
x [*,*]
:      1      2      3      4      :=
1      0      0      0      1
2      0      0      0      1
3      0      0      0      1
4      0      0      0      1
5      0      0      0      1
6      1      0      0      0
7      0      1      0      0
8      0      1      0      0
9      1      0      0      0
10     0      0      1      0
11     0      1      0      0
12     0      1      0      0
13     0      1      0      0
14     0      1      0      0
15     0      0      0      1
16     0      1      0      0
;
```